

## Estimating Ordinal Reliability Using SAS®

Laura Ring Kapitula, Grand Valley State University

### ABSTRACT

In evaluation instruments and tests individual items are often collected using an ordinal measurement or likert type scale. Typically measures such as Cronbach's alpha are estimated using the standard Pearson correlation. Gadderman, Guhn and Zumbo (2012) illustrate how using the standard Pearson correlations may yield biased estimates of reliability when the data are ordinal and present methodology for using the polychoric correlation in reliability estimates as an alternative. This session shows how to implement the methods of Gadderman, Guhn and Zumbo using SAS® software. An example, will be presented that incorporates these methods in the estimation of the reliability of an active learning post-occupancy evaluation instrument developed by Steelcase Education Solutions researchers.

### INTRODUCTION

In practice Cronbach's alpha is often used to estimate reliability. The typical method of computing Cronbach's alpha in SAS® is using the CORR procedure and the ALPHA option. This default method calculates alpha based on the Pearson correlation matrix (or standard covariance matrix) and has the underlying assumption that the data are continuous. When the data are ordinal this method may give biased results (Zumbo, Gadermann and Zeisser, 2007). Gadderman, Guhn and Zumbo (2012) provide code to estimate reliability for ordinal data using R, however it was the desire of the author to be able to more easily automate reporting and do the whole analysis use SAS® solutions. In this paper we discuss and give SAS® code to estimate reliability for ordinal data.

### METHODS

Zumbo, Gadermann and Zeisser (2007) carried out a simulation study that illustrated negative bias in Cronbach's alpha as computed on the Pearson correlation when the data are ordinal or likert type and they suggest two alternatives: ordinal alpha and ordinal theta. Both ordinal alpha and ordinal theta are calculated using the polychoric correlation.

Drasgow (1986) provides an overview of polychoric correlation. The basic idea is that for each of the two variables there is an underlying latent normal random variable but all that is observed is the interval a variable belongs in with unknown end points. It is assumed the ordinal variable with k levels divides the data into k intervals and using maximum likelihood estimation the underlying correlation between the latent variables is estimated. That estimate is called the polychoric correlation.

Gadderman, Guhn and Zumbo (2012) point out that once the polychoric correlation matrix is obtained it can be used to estimate measures of reliability and for factor analysis. The ordinal alpha can be calculated using the correlation matrix as :

$$\alpha = \frac{p}{p-1} \left( 1 - \frac{p}{p + \sum_{i \neq j} r_{ij}} \right),$$

where  $p$  is the number of items and  $r_{ij}$  is the polychoric correlation between items  $i$  and  $j$ . Note that  $p + \sum_{i \neq j} r_{ij}$  is just the sum of all values in the correlation matrix, since in a correlation matrix the diagonal elements are all 1, they sum to be the number of elements and the double sum adds up all off diagonal elements (the correlations), note that each one would be repeated twice given the symmetry in the matrix.

The other measure suggested in Zumbo, Gadermann and Zeisser (2007) is the reliability estimate Coefficient theta proposed by Armor (1974) that is based on a principal components analysis. If the single factor solution is reasonable for the items, then

$$\theta = \frac{p}{p-1} \left( 1 - \frac{1}{\lambda_1} \right),$$

Where  $\lambda_1$  is the largest eigenvalue from the principal component analysis of the correlation matrix for the items.

Using the FREQ procedure in SAS® 9.3 (and in SAS® 9.4 the CORR procedure) the polychoric correlation matrix can be output to a data set and read into the IML procedure to compute ordinal alpha and ordinal theta.

In addition to looking at measures of inter-item consistency it is helpful to carry out a factor analysis to see if there is evidence that the single factor solution is reasonable for the items, a property that is often called unidimensionality. The polychoric correlation matrix can be used to examine this property in the same manner one would use the Pearson correlation matrix if the data were continuous (Gadderman, Guhn and Zumbo, 2012). Finally, if there is evidence that the reliability is high and the instrument is unidimensional with fairly equal factor loadings the items may be summed to get an overall measure. Further analysis can be done to check that no items have exceptionally low or high correlations with the overall measure. Typically the number of items are sufficient that it is reasonable to treat the overall summed measure as continuous, given this is the case the polyserial correlation is an appropriate measure of the association between the ordinal individual items and the overall measure. The polyserial correlation is similar to the polychoric correlation but only one of the variables is treated as measuring an underlying latent variable and the other is measured directly( Drasgow ,1986). The polyserial correlation can be obtained using the CORR procedure in SAS<sup>®</sup> 9.3.

## APPLIED ILLUSTRATION

An active learning post-occupancy evaluation (AL-POE) instrument was developed by Steelcase Education Solutions to measure how the built environment impacts classroom engagement. For details of this work see Scott-Webber, Strickland, and Kapitula(2013).

The instrument was designed to measure four different conditions, a learning practices condition pre- and post- test and a solutions condition pre- and post- test. The reliability and item analysis was done on the pre- and post-test responses for learning practices items, and the pre- and post-test responses for solutions items separately. Each of these four conditions had twelve AL-POE items. Here we will illustrate the method with the pre-solutions condition and then summarize the results for all four of the conditions on a sample of data collected from college students in classroom environments designed to optimize active learning.

The Pearson correlations can be calculated with the CORR procedure in SAS<sup>®</sup> and the ALPHA option gives standard reliability measures calculated using the Pearson correlations. Saving the correlation matrix and then printing it out gives more control over how the matrix is presented. See the appendix for how to code this in SAS<sup>®</sup>.

Pearson Correlations for Pre-test Solutions Section (n=268)												
item	q1	q2	q3	q4	q5	q6	q7	q8	q9	q10	q11	q12
q1	1.000	0.586	0.653	0.582	0.482	0.435	0.424	0.607	0.522	0.567	0.527	0.623
q2	0.586	1.000	0.663	0.596	0.482	0.441	0.538	0.656	0.436	0.657	0.588	0.662
q3	0.653	0.663	1.000	0.635	0.458	0.403	0.476	0.597	0.458	0.523	0.534	0.604
q4	0.582	0.596	0.635	1.000	0.553	0.401	0.496	0.592	0.526	0.539	0.532	0.611
q5	0.482	0.482	0.458	0.553	1.000	0.511	0.431	0.487	0.433	0.516	0.388	0.537
q6	0.435	0.441	0.403	0.401	0.511	1.000	0.369	0.513	0.446	0.528	0.446	0.555
q7	0.424	0.538	0.476	0.496	0.431	0.369	1.000	0.538	0.458	0.500	0.466	0.494
q8	0.607	0.656	0.597	0.592	0.487	0.513	0.538	1.000	0.598	0.666	0.577	0.760
q9	0.522	0.436	0.458	0.526	0.433	0.446	0.458	0.598	1.000	0.566	0.408	0.523
q10	0.567	0.657	0.523	0.539	0.516	0.528	0.500	0.666	0.566	1.000	0.629	0.721
q11	0.527	0.588	0.534	0.532	0.388	0.446	0.466	0.577	0.408	0.629	1.000	0.647
q12	0.623	0.662	0.604	0.611	0.537	0.555	0.494	0.760	0.523	0.721	0.647	1.000

**Table 1. Pearson Correlations for Pre-test Solutions Section**

The polychoric correlation can be obtained using PROC FREQ, then the output data can be manipulated to take the form of a correlation matrix that can be read into other procedures such as PROC FACTOR and PROC IML<sup>1</sup>.

<sup>1</sup> This section of code was inspired by the very helpful SAS FAQ:  
*How do I compute tetrachoric/polychoric correlations in SAS?* UCLA: Statistical Consulting Group.  
 from <http://www.ats.ucla.edu/stat/sas/faq/tetrac.htm> (accessed February 21, 2014).


Polychoric correlations for Pre-test Solutions Section(n=268)												
item	q1	q2	q3	q4	q5	q6	q7	q8	q9	q10	q11	q12
q1	1.000	0.660	0.724	0.652	0.539	0.517	0.495	0.691	0.599	0.676	0.618	0.716
q2	0.660	1.000	0.737	0.668	0.539	0.516	0.595	0.732	0.504	0.730	0.671	0.732
q3	0.724	0.737	1.000	0.703	0.512	0.488	0.549	0.689	0.545	0.628	0.624	0.689
q4	0.652	0.668	0.703	1.000	0.618	0.470	0.561	0.663	0.609	0.615	0.600	0.687
q5	0.539	0.539	0.512	0.618	1.000	0.544	0.495	0.545	0.496	0.566	0.461	0.579
q6	0.517	0.516	0.488	0.470	0.544	1.000	0.432	0.602	0.509	0.574	0.541	0.619
q7	0.495	0.595	0.549	0.561	0.495	0.432	1.000	0.591	0.525	0.552	0.521	0.548
q8	0.691	0.732	0.689	0.663	0.545	0.602	0.591	1.000	0.665	0.736	0.631	0.823
q9	0.599	0.504	0.545	0.609	0.496	0.509	0.525	0.665	1.000	0.637	0.485	0.580
q10	0.676	0.730	0.628	0.615	0.566	0.574	0.552	0.736	0.637	1.000	0.691	0.778
q11	0.618	0.671	0.624	0.600	0.461	0.541	0.521	0.631	0.485	0.691	1.000	0.731
q12	0.716	0.732	0.689	0.687	0.579	0.619	0.548	0.823	0.580	0.778	0.731	1.000

**Table 2. Polychoric Correlations for Pre-test Solutions Section**

Notice that the polychoric correlations tend to be higher than the Pearson correlations.

After the polychoric correlation is saved to a data set the reliability measures can be found using PROC IML. Matrix computations allow them to be calculated with very little code. The IML code is included as part of the CALCALPHA macro and is given in the appendix. For these data the standard alpha is  $\alpha = 0.9209$  and the ordinal alpha is  $\alpha = 0.9361$ . Given the standard alpha tends to be biased downward; this result is consistent with what would be expected. The standard theta is  $\theta = 0.9215$  and the ordinal theta is  $\theta = 0.9367$ . All measures indicate a high degree of internal item consistency.

A factor analysis was carried out using the polychoric correlation matrix. As is indicated by the scree plot below there is evidence of one underlying factor.



**Figure 1. Plots from Factor Analysis of the Polychoric Correlation Matrix**

About 96% of the variation is explained by the first factor and this and the plots above provide evidence for unidimensionality. The sample size is so large that the plots and proportion of variance are more informative than looking at hypothesis tests for assessing if the one factor model is sufficient.

Finally, the polyserial correlations were found between the overall composite measure and each of the items using PROC CORR. Polyserial correlation between the overall composite measure and individual items ranged from 0.71 and 0.90 for these data.

Similar analyze were done on each of the four conditions. The ordinal alphas were all greater than 0.91 and the factor analysis indicated the instrument was unidimensional in all four conditions. These results gave support for creating composite variables for each condition, a pre-practices composite variable, a post-practices composite variable, a pre-solutions composite variable and a post-practices composite variable. The composite variables were formed by summing the individual responses for the twelve items for each condition to create a scale with a minimum value of zero and a maximum value of 48. Item analysis indicated that the degree of correlation between each individual item response in a condition and the composite score for that condition were all greater than 0.6.

## CONCLUSIONS

If the ordinality of measurement in an instrument is ignored it is possible that reliability estimates will be biased negatively. This paper illustrates how to use SAS<sup>®</sup> to calculate reliability measures and perform factor analysis on the polychoric correlations and obtain more accurate estimates of the reliability of an instrument when the items are measured using Likert type scales. It is preferred to estimate these values for each new implementation of a survey instrument and coding them in SAS<sup>®</sup> allows these values to easily be incorporated as part of a generated report.

## APPENDIX

### Source Code

```
/* prepare data, itemdata contains the responses for the items in the instrument*/
data itemdata;
  set itemdata;
  if nmiss(of q1 -q12) =0; /* use listwise deletion*/
  totscore= sum(of q1-q12);
  label totscore = "Overall Score";
  keep totscore q1-q12;
run;

/* Calculate and Store Pearson Correlation Matrix and obtain
the standard reliability estimates. */

proc corr data=itemdata outp=pearsonc alpha noprob nosimple;
  var q1-q12;
run;

proc print data=pearsonc(where=( _type_="CORR")) noobs label ;
var _name_ q1-q12;
format q1-q12 5.3;
label _name_="item";
title "Pearson Correlation Matrix for Post-Solutions";
run;

ods rtf exclude all; /* keeps a lot of output from being written to the report*/
/* Calculate the Polychoric Correlation and Save it into a Correlation Data set*/
/* http://www.ats.ucla.edu/stat/sas/faq/tetrac.htm was used
in the writing of this code*/
proc freq data=itemdata ;
  tables (q1-q12)*(q1-q12) /plcorr noprint ;
  ods output measures=mycorr (where=(statistic="Tetrachoric Correlation"
                                or statistic="Polychoric Correlation")
                                keep = statistic table value);

run;
ods rtf exclude none;

data mycorr;
  set mycorr;
  x = scan(table, 2, " *");
  y = scan(table, 3, " *");
  group=substr(y,2)+0;
run;
proc sort data=mycorr;
  by group x;
```

```

proc transpose data = mycorr out=corrmatrix (drop = group _name_) ;
id x;
by group;
var value ;
run;
title "Polychoric Correlation Matrix";
data corrmatrix(type=corr);
  _TYPE_='CORR';
  set corrmatrix end=endc;
  _name_=compress("q"||put(_n_,2.));
output;
array q _numeric_;
if endc then do;
  _TYPE_='N';
  do over q;
    q=268;
  end;
output;
end;
run;
proc print data = corrmatrix(where=( _type_="CORR")) noobs label;
var _name_ q1 q2 q3 q4 q5 q6 q7 q8 q9 q10 q11 q12;
label _name_ ="item";
format q1-q12 5.3;
run;
/* do Factor Analysis on the Correlation Matrix */
proc factor data=corrmatrix method=ml nfactors=1 plots=all;
run;
/* obtain polyserial correlations between items and the overall composite score*/
proc corr data=itemdata polyserial;
var totscore;
with q1-q12;

run;

/* calculate alpha and omega and store it in a data set */
%macro varlist;
  %do i = 1 %to 12;
    q&i
  %end;
%mend;

%macro calcalpha(qs=q, cordat=corrmatrix, out=alpha);
/* qs= the subsection of the instrument */
/*cordat is the correlation matrix data */
proc iml;
  use &cordat(where=( _type_="CORR")) var {%varlist};
  read all var {%varlist } into itemcor;
  section="&qs";
  datused="&cordat";
  numbitem=trace(itemcor);
  one=j(numbitem,1);
  print 'The Input Correlation';
  print itemcor;
  print 'Number of items' numbitem;
  sumcor=(one`)*itemcor*one; /*sum all correlations*/
  alpha=(numbitem/(numbitem-1))*(1-(numbitem / sumcor));
  print alpha;
  L = eigval(itemcor);
  theta=(numbitem/(numbitem-1))*(1-1/L[1]);
  print 'Theta';
  print theta;

```

```

/* to save results in a data set */
create &out var {section datused alpha theta}; /** create data set **/
append; /** write data **/
close &out; /** close the data set **/

quit;
run;
%mend;

/* call the macro */
%calcalpha(qs=q, cordat=corrmatrix);
%calcalpha(qs=q, cordat=pearsonc, out=alphap);

```

## REFERENCES

- Armor, D. J. (1974). Theta reliability and factor scaling. In H. Costner (Ed.), *Sociological methodology* (pp. 17-50). San Francisco: Jossey-Bass.
- Drasgow, F. (1986) Polychoric and polyserial correlations. Pp. 68-74 in S. Kotz and N. Johnson, eds., *The Encyclopedia of Statistics, Volume 7*. Wiley.
- Gadermann, A.M., Guhn, M. and Zumbo, B.D. (2012). Estimating ordinal reliability for Likert-type and ordinal item response data: A conceptual, empirical, and practical guide. *Practical Assessment, Research and Evaluation*, 17(3). Retrieved Sept. 25, 2012 from <http://pareonline.net/pdf/v17n3.pdf>
- Scott-Webber, L., Strickland, A. and Kapitula, L.R. (2013). Built Environments Impact Behaviors Results of an Active Learning Post-Occupancy Evaluation. *Planning for Higher Education Journal*, V42N1. Retrieved February 21, 2014 from [http://www.steelcase.com/en/products/category/educational/case-studies/documents/phev42n1\\_article\\_built-environments.pdf](http://www.steelcase.com/en/products/category/educational/case-studies/documents/phev42n1_article_built-environments.pdf)
- Zumbo, B. D., Gadermann, A. M., and Zeisser, C.. (2007). Ordinal Versions of Coefficients Alpha and Theta For Likert Rating Scales. *Journal of Modern Applied Statistical Methods*, 6, 21-29.

## ACKNOWLEDGMENTS

I would like to thank Steelcase Education Solutions and specifically Lennie Scott-Webber and Aileen Strickland, who developed the AL-POE instrument and provided me the opportunity to work with them on the AL-POE project. I would also like to thank the SAS Institute for providing me with a SAS® Global Forum Faculty Scholarship and the Center for Scholarly and Creative Excellence and the Department of Statistics at GVSU for providing me with travel expenses to SAS® global.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Laura Ring Kapitula, PhD  
 Department of Statistics Grand Valley State University  
 A-1-178 Mackinac Hall  
 1 Campus Drive  
 Allendale, MI 49401  
 616-331-8019  
 kapitull@gvsu.edu  
[http://www.sascommunity.org/wiki/Estimating\\_Ordinal\\_Reliability\\_Using\\_SAS%C2%AE](http://www.sascommunity.org/wiki/Estimating_Ordinal_Reliability_Using_SAS%C2%AE)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.